UMass Boston Computer Science

CS450 High Level Languages

Implementing Function Calls

Thursday, April 22, 2025

Logistics

- HW 10 in
  - ~~due: Tues 4/22 11am EST~~
- HW 11 out
  - due: Tues 4/29 11am EST

Last Time

The “CS450” Programming Lang!

Programmer writes:

```
;; A Program is one of:
;; - Atom
;; - `(+ ,Program ,Program)
;; - `(× ,Program ,Program)
;; - `(~= ,Program ,Program)
;; - `(iffy ,Program ,Program ,Program)
```

Last Time

Adding Variables

Programmer writes:

```
;; A Program is
;; - Atom
;; - Variable
;; - ...
```

```
;; An Variable is a:
;; - Symbol
```

Q_1: What is the "meaning" of a variable?

A_1: Whatever "value" it represents

Q_2: Where do these "values" come from?

A_2: Other parts of the program!

```
;; An AST is one of:
;; - …
;; - (mk-var Symbol)
;; …

;; …
(struct vari [name])
```

Hint: Don't use "**var**" (reserved for a Racket **match** pattern) for struct name

```
;; A Result is one of:
;; - Boolean
;; - String
;; - ???
```

"meaning" of the program

The **run** function needs to "remember" these values (with an **accumulator**!)

run

(JS semantics)

Guess Who's Back!

HW11

Design Recipe For Accumulator Functions

When a function needs **"extra information":**

1. *Specify* **accumulator:**
   - Name
   - Signature
   - Invariant
2. *Define* internal "helper" fn with extra **accumulator** arg

(Helper fn does not need extra description, statement, or examples, if they are the same …)

3. *Call* "helper" fn , with initial accumulator value, from original fn

run, with an accumulator

```
;; run: AST -> Result
;; Computes result of running a CS450 Lang program AST
(define (run p)
  ;; accumulator acc : Environment
  ;; invariant: Contains variable values … currently in-scope
  (define (run/acc p acc)
    (match p
      [(num n) n]
      [(add x y) (450+ (run/acc x) (run/acc y))]))
 (run/acc p  ???  ))
```

Environments

- A **data structure** that **“associates” two things** (`var`, `val`) **together**
  - E.g., **maps**, **hashes**, **list-of-pairs**



Environments

- A **data structure** that **"associates" two things** (`var`, `val`) **together**
  - E.g., **`maps`**, **`hashes`**, **list-of-pairs**

```
;; An Environment is one of:
;; - empty
;; - (cons (list Var Result) Environment)
```

- Needed operations:
  - `env-add    : Env Var Result -> Env`
  - `env-lookup : Env Var -> Result`

Environments

```
;; An Environment is one of:
;; - empty
;; - (cons (list Var Result) Environment)
```

```
;; interpretation: a runtime environment
;; gives meaning to cs450lang variables
```

```
;; for duplicates, vars at front of
;; list shadow those in back
```

- Needed operations:
  - `env-add    : Env Var Result -> Env`
  - `env-lookup : Env Var -> Result`

Think about examples where this happens!

`env-add` examples

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; env-add: Env Var Result -> Env
```

```
(check-equal? (env-add '() 'x 1)
              '((x 1)) )  ; add to empty
```

```
(check-equal? (env-add '((x 1)) 'y 2)
              '((y 2) (x 1)) ) ; add new var
```

```
(check-equal? (env-add '((x 1)) 'x 3)
              '((x 3) (x 1)) ) ; add shadowed var
```

```
;; for duplicates, vars at front of
;; list shadow those in back
```

Env template

2 cases

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

2 cases

```
(define (env-fn env … )
  (cond
   [(empty? env) … ]
   [else
    (match-let
        ([(cons (list x result) rest-env) env])
      … x … result … (env-fn rest-env … ) … ]))
```

2nd case extracts components of compound data

Env template

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
(define (env-fn env … )
  (cond
   [(empty? env) … ]
   [else
    (match-let
        ([(cons (list x result) rest-env) env])
        ([`((,x ,result) . ,rest-env) env])
      … x … result … (env-fn rest-env … ) … ]))
```

cons pattern

Quasiquote pattern

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; env-add: Env Var Result -> Env
(define (env-add env new-x new-res)
  (cond
   [(empty? env) … ]
   [else
    (match-let
        ([(cons (list x result) rest-env) env])
        ([`((,x ,result) . ,rest-env) env])
      … x … result …(env-add rest-env … ) … ]))
```

Start with template

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; env-add: Env Var Result -> Env
(define (env-add env new-x new-res)
  (cond
   [(empty? env) ...]
   [else   … ]))
```

Examples

```
(check-equal? (env-add '() 'x 1)
              '((x 1)) ) ; add to empty
```

Base case – empty env

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; env-add: Env Var Result -> Env
(define (env-add env new-x new-res)
  (cond
   [(empty? env) (cons (list new-x new-res) env)]
   [else   … ]))
```

Base case – empty env

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; env-add: Env Var Result -> Env
(define (env-add env new-x new-res)
  (cond
   [(empty? env) (cons (list new-x new-res) env)]
   [else                ...]))
```

recursive case?
(non-empty env)

Examples

```
(check-equal? (env-add '((x 1)) 'y 2)
              '((y 2) (x 1)) ) ; add new var
```

```
(check-equal? (env-add '((x 1)) 'x 3)
              '((x 3) (x 1)) ) ; add shadowed var
```

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; env-add: Env Var Result -> Env
(define (env-add env new-x new-res)
  (cond
   [(empty? env) (cons (list new-x new-res) env)]
   [else         (cons (list new-x new-res) env)]))
```

recursive case?
(non-empty env)

Sometimes you start with template … but don't use all the pieces!

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; env-add: Env Var Result -> Env
(define (env-add env new-x new-res)
  (cons (list new-x new-res) env))
```

Collapse similar cases

env-lookup examples

```
;; env-lookup: Env Var -> Result
```

```
(check-equal? (env-lookup '((y 2) (x 1)) 'x)
              1) ; no dup
```

```
(check-equal? (env-lookup '((x 2) (x 1)) 'x)
              2) ; duplicate
```

```
(check-equal? (env-lookup '() 'x)
              UNDEFINED-ERROR ) ; not found!
```

```
;; A Result is one of:
;; - Number
;; …
;; - UNDEFINED-ERROR
```

An "error" is a valid program "Result"!

… for now, just represent with special `Result` value

NOTE: we don't want Racket exception because this is a "CS450 Lang error" … Racket program runs fine!

env-lookup

```
;; env-lookup: Env Var -> Result
(define (env-lookup env target-x)
  (cond
   [(empty? env) … ]
   [else
    (match-let
        ([`((,x ,res) . ,rest-env) env])
      … x … res … (env-lookup rest-env … ) … ]))
```

TEMPLATE!

env-lookup: empty (error) case

```
;; env-lookup: Env Var -> Result
(define (env-lookup env target-x)
  (cond
   [(empty? env) UNDEFINED-ERROR]
   [else
    … ]))
```

env-lookup: non-empty case

```
;; env-lookup: Env Var -> Result
(define (env-lookup env target-x)
  (cond
   [(empty? env) UNDEFINED-ERROR]
   [else                                  Extract the pieces
    (match-let
        ([`((,x ,res) . ,rest-env) env])
      … x … res … (env-lookup rest-env … ) … ]))
```

env-lookup: non-empty case

```
;; env-lookup: Env Var -> Result
(define (env-lookup env target-x)
  (cond
   [(empty? env) UNDEFINED-ERROR]
   [else
    (match-let
        ([`((,x ,res) . ,rest-env) env])
      (if (var=? x target-x)
          res
          … (env-lookup rest-env … ) … ]))
```

Found target-x

env-lookup: non-empty case

```
;; env-lookup: Env Var -> Result
(define (env-lookup env target-x)
  (cond
   [(empty? env) UNDEFINED-ERROR]
   [else
    (match-let
        ([`((,x ,res) . ,rest-env) env])
      (if (var=? x target-x)
          res
          (env-lookup rest-env target-x)))]))
```

Else, recursive call with remaining env

run, with an Environment accumulator

```
;; run: AST -> Result
(define (run p)
  ;; accumulator env : Environment
  ;; invariant: contains in-scope var + results
  (define (run/env p env)
    (match p
     [(num n) n]
     [(add x y) (450+ (run/env x) (run/env y))]))
 (run/env p  ???  ))
```

run, with an Environment accumulator

```
;; run: AST -> Result
(define (run p)
  ;; accumulator env : Environment
  ;; invariant: contains in-scope var + results
  (define (run/env p env)
    (match p
      …
      [(vari x) (env-lookup env x)]
      [(bind x e body) … (env-add env x (run/env e env)) …]
      …   ))
 (run/env p  ???  ))
```

Don't name this `var`

run, with an Environment accumulator

TODO:
- When are variables “added” to environment**???**
- What is initial environment?

```
;; run: AST -> Result
(define (run p)
  ;; accumulator env : Environment
  ;; invariant: contains in-scope var + results
  (define (run/env p env)
    (match p
      …
      [(vari x) (env-lookup env x)]
      [(bind ??? body) … (env-add env x ??? e env)) …]
      …  ))
 (run/env p ???  ))
```

Programs that Add Variables to Environment

```
;; A Program is one of:
;; - Atom
;; - Variable (Var)
;; - ?????
```

Programs that Add Variables to Environment

```
;; A Program is one of:
;; - Atom
;; - Variable (Var)
;; - `(bind [,Var ,Program] ,Program)
;; - ...
```

(like "`let`" in other langs)

Programs that Add Variables to Environment

```
;; A Program is one of:
;; - Atom
;; - Variable (Var)
;; - `(bind [,Var ,Program] ,Program)
;; - ...
```

parse

```
;; An AST is one of:
;; - …
;; - (mk-var Symbol)
;; - (mk-bind Symbol AST AST)
;; - …

;; …
(struct vari [name])
(struct bind [var expr body])
;; …
```

Need to be more careful parsing

Valid Program?

‘(bind)

‘(bind [])

‘(bind [1 2] 3)

Interlude: Racket exceptions

Exceptions are just special structs

Super struct (enables using **exception API**)

```
(struct exn:fail:syntax:cs450 exn:fail:syntax [])
```

```
(define/contract (parse p)
  (-> Program? AST?)
  (match p
   [(? atom?) (parse-atom p)]
     ...
   [`(,fn . ,args) ... ]
   [_ (error … )]))
```

Interlude: Racket exceptions

Exceptions are just special structs

Super struct (enables using exception API)

```
(struct exn:fail:syntax:cs450 exn:fail:syntax [])
```

```
(define/contract (parse p)
  (-> Program? AST?)
  (match p
   [(? atom?) (parse-atom p)]
     ...
   [`(,fn . ,args) ... ]
   [_ (raise-syntax-error
        'parse "not a valid CS450 Lang program" p
       #:exn exn:fail:syntax:cs450)]))
```

```
(define/contract (parse p)
  (-> Program? AST?)
  (match p
        ...
   [`(bind [,(and (? symbol?) x) ,e] ,bod) ... ]

        ...

   [`(,fn . ,args) ... ]
   [_ (raise-syntax-error
         'parse "not a valid CS450 Lang program" p
         #:exn exn:fail:syntax:cs450)]))
```

```
(define/contract (parse p)
  (-> Program? AST?)
  (match p
        ...
   [`(bind [,(and (? symbol?) x) ,e] ,bod) ... ]
   [`(bind . _)
     (raise-syntax-error 'parse “invalid bind syntax" p
        #:exn exn:fail:syntax:cs450)   ]
   [`(,fn . ,args) ... ]
   [_ (raise-syntax-error
        'parse "not a valid CS450 Lang program" p
        #:exn exn:fail:syntax:cs450)]))
```

Bind parse error case

Programs that Add Variables to Environment

```
;; A Program is one of:
;; - Atom
;; - Variable (Var)
;; - `(bind [,Var ,Program] ,Program)
;; - ...
```

parse

```
;; An AST is one of:
;; - …
;; - (mk-var Symbol)
;; - (mk-bind Symbol AST AST)
;; - …

;; …
(struct vari [name])
(struct bind [var expr body])
;; …
```

Need to be more careful parsing

Valid Program?

```
(check-exn exn:fail:syntax:cs450?
     (λ () (eval450 '(bind)) ))
```

```
(check-exn exn:fail:syntax:cs450?
     (λ () (eval450 '(bind [])) ))
```

```
(check-exn exn:fail:syntax:cs450?
     (λ () (eval450 '(bind [1 2] 3)) ))
```

Programs that Add Variables to Environment

```
;; A Program is one of:
;; - Atom
;; - Variable (Var)
;; - `(bind [,Var ,Program] ,Program)
;; - ...
```

parse

```
;; An AST is one of:
;; - …
;; - (mk-var Symbol)
;; - (mk-bind Symbol AST AST)
;; - …

;; …
(struct vari [name])
(struct bind [var expr body])
;; …
```

run

???

Bind scoping examples

```
;; A Program is one of:
;; - Atom
;; - Variable (Var)
;; - `(bind [,Var ,Program] ,Program)
;; - ...
```

bind obeys "lexical" or "static" scoping

Generally accepted to be "best choice" for programming language design
(bc it's determined only by program syntax)

run, with **bind**

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
;; run: AST -> Result
(define (run p)
  ;; accumulator env : Environment
  ;; invariant: contains in-scope var + results
                                    env)
```

```
; An AST is one of:
; - …
; - (mk-bind Symbol AST AST)
```

```
      [(vari x) (env-lookup env x)]
      [(bind x e body) … (env-add env x (run/env e env)) …]
       …    ))
 (run/env p  ???  ))
```

Environment has **Results** (not **AST**)

How to convert **AST** to **Result**?

(From template!)

Add to environment

Be careful to get correct "**scoping**" (x not visible in expression e, so use unmodified input env)

`run`, with **`bind`**

`run` must produce **`Result`**

```
;; run: AST -> Result
(define (run p)
  ;; accumulator env : Environment
  ;; invariant: contains in-scope var + results
  (define (run/env e env)
    (type-case AST e
      [(vari x) (env-lookup env x)]
      [(bind x e body) ??? (env-add env x (run/env e env)) …]
      …  ))
  (run/env p  ???  ))
```

```
; An AST is one of:
; - …
; - (mk-bind Symbol AST AST)
```

run, with **bind**

```
;; run: AST -> Result
(define (run p)
  ;; accumulator env : Environment
  ;; invariant: contains in-scope var + results
  (define (run/env p env)
    (match p
      …
     [(vari x) (env-lookup env x)]
     [(bind x e body) (run/env body (env-add env x (run/env e env)))]
      …   ))
 (run/env p  ???  ))
```

(From template!)

run body with new env containing **x**

Initial Environment?

TODO:
- ~~When are variables "added" to environment~~
- What is initial environment? `empty` (for now)

```
;; run: AST -> Result
(define (run p)
  ;; accumulator env : Environment
  ;; invariant: contains in-scope var + results
  (define (run/env p env)
    (match p
      …
      [(vari x) (env-lookup env x)]
      [(bind x e body) (run/env body (env-add env x (run/env e env)))]
      …    ))
 (run/env p  ???  ))
```

`empty` ??? (for now)

Initial Environment

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(+ ,Program ,Program)
;; - `(× ,Program ,Program)
```

These don't need to be separate constructs

Put these into "initial" environment

Initial Environment

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(+ ,Program ,Program)
;; - `(× ,Program ,Program)
```

Put these into “initial” environment

```
;; An Environment (Env) is one of:
;; - empty
;; - (cons (list Var Result) Env)
```

```
(define INIT-ENV
`((+ ,450+)
  (× ,450*)))
```

+ variable

Maps to our “450+” function

New kind of Result

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - (Racket) Function
```

For Program: +

Initial Environment

How do users call these functions???

```
(define INIT-ENV `((+ ,450+) (× ,450*)))
```

```
(define (run p)

  ;; accumulator env : Environment
  (define (run/e p env)
    (match p
      …
     [(vari x) (lookup env x)]
     [(bind x e body) (run/e body (env-add env x (run/e e env)))]
      … ))
 (run/e p INIT-ENV ))
```

Function Application in CS450 Lang

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(fncall ,Program . ,List<Program>)
```

function

arguments

“rest” arg

Allows arbitrary number of args

Function Application in CS450 Lang: Examples

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(fncall ,Program . ,List<Program>)
```

function

arguments

```
(fncall + 1 2)
```

Programmers shouldn’t need to write the explicit “fncall”

Function Application in CS450 Lang: Examples

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(,Program . ,List<Program>)
```

```
(+ 1 2)
```

Function Application in CS450 Lang: Examples

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(,Program . ,List<Program>)
;; - (cons Program List<Program>)
```

Can go back to (simpler?) **cons** notation

```
(+ 1 2)
```

Must be careful when parsing this

Function call case (must be last, why?)

```
(bind 1 2)
```

(should not be function call)

Function Application in CS450 Lang

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(,Program . ,List<Program>)
;; - (cons Program List<Program>)
```

parse

```
;; An AST is one of:
;; - …
;; - (mk-var Symbol)
;; - (mk-bind Symbol AST AST)
;; - (mk-call AST List<AST>)

(struct vari [name])
(struct bind [var expr body])
(struct call [fn args])
```

“Running” Function Calls

TEMPLATE: extract pieces of compound data

```
;; An AST is one of:
;; - …
;; - (mk-var Symbol)
;; - (mk-bind Symbol AST AST)
;; - (mk-call AST List<AST>)

(struct vari [name])
(struct bind [var expr body])
(struct call [fn args])
```

```
(define (run p)


  (define (run/e p env)
    (match p
           …
      [(call fn args) (apply
                       (run/e fn env)
                       (map (curryr run/e env) args))]

           …
      ))
  (run/e p INIT-ENV))
```

“Running” Function Calls

```
;; An AST is one of:
;; - …
;; - (mk-var Symbol)
;; - (mk-bind Symbol AST AST)
;; - (mk-call AST List<AST>)
```

```
(define (run p)

  (define (run/e p env)
    (match p
           …
      [(call fn args) (apply
                       (run/e fn env)
                       (map (curry ??? run/e env) args))]

           …
      ))
  (run/e p INIT-ENV))
```

TEMPLATE: recursive calls

“run” args before calling function – “call by value”

“Running” Function Calls

How do we actually run the function?

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - (Racket) Function
```

```
(define (run p)


  (define (run/e p env)
    (match p
          …
      [(call fn args) (apply
                       (run/e fn env)
                       (map (curryr run/e env) args))]

          …
      ))
  (run/e p INIT-ENV))
```

Calls racket fns

???

Does this work???

“Running” Non-Functions

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - (Racket) Function
```

```
(define (run p)

  (define (run/e p env)
    (match p
           …
      [(call fn args) (apply
                       (run/e fn env)
                       (map (curryr run/e env) args))]
           …
      ))
  (run/e p INIT-ENV))
```

Example: `(eval450 '(10 10)) ; apply non-fn`

“Running” Non-Functions

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - NON-FUNCTION-ERROR
;; - (Racket) Function
```

```
(define (run p)

  (define (run/e p env)
    (match p
           …
      [(call fn args) (450apply
                       (run/e fn env)
                       (map (curryr run/e env) args))]
           …
      ))
  (run/e p INIT-ENV))
```

Function application for CS450 Lang

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - NON-FUNCTION-ERROR
;; - (Racket) Function
```

```
;; 450apply : Result Listof<Result> -> Result
```

```
(define (450apply fn args)
  (cond
   [(number? fn)                ... ]
   [(UNDEFINED-ERROR? fn)       ... ]
   [(NON-FUNCTION-ERROR? fn)    ... ]
   [(procedure? fn)             ... ]))
```

TEMPLATE?

Function application for CS450 Lang

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - NON-FUNCTION-ERROR
;; - (Racket) Function
```

```
;; 450apply : Result Listof<Result> -> Result
```

```
(define (450apply fn args)
  (cond
   [(number? fn) NON-FUNCTION-ERROR]
   [(UNDEFINED-ERROR? fn)     ... ]
   [(NON-FUNCTION-ERROR? fn) ... ]
   [(procedure? fn)           ... ]))
```

Function application for CS450 Lang

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - NON-FUNCTION-ERROR
;; - (Racket) Function
```

```
;; 450apply : Result Listof<Result> -> Result
```

```
(define (450apply fn args)
  (cond
   [(number? fn) NON-FUNCTION-ERROR]
   [(UNDEFINED-ERROR? fn)     ... ]
   [(NON-FUNCTION-ERROR? fn) ... ]
   [(procedure? fn) (apply fn args)]))
```

Now this works

Function application for CS450 Lang

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - NON-FUNCTION-ERROR
;; - (Racket) Function
```

```
;; 450apply : Result Listof<Result> -> Result
```

```
(define (450apply fn args)
  (cond
   [(number? fn) NON-FUNCTION-ERROR]
   [(UNDEFINED-ERROR? fn)     ... ]
   [(NON-FUNCTION-ERROR? fn) NON-FUNCTION-ERROR]
   [(procedure? fn) (apply fn args)]))
```

Function application for CS450 Lang

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - NON-FUNCTION-ERROR
;; - (Racket) Function
```

```
;; 450apply : Result Listof<Result> -> Result
```

```
(define (450apply fn args)
  (cond
    [(number? fn) NON-FUNCTION-ERROR]
    [(UNDEFINED-ERROR? fn) UNDEFINED-ERROR]
    [(NON-FUNCTION-ERROR? fn) NON-FUNCTION-ERROR]
    [(procedure? fn) (apply fn args)]))
```

UNDEFINED should have precedence over NON-FN-ERR

Function application for CS450 Lang

```
;; A Result is one of:
;; - Number
;; - UNDEFINED-ERROR
;; - NON-FUNCTION-ERROR
;; - (Racket) Function
```

Add ARITY-ERROR ???

```
;; 450apply : Result Listof<Result> -> Result
```

For now, we only use variable-arity functions

```
(define (450apply fn args)
  (cond
   [(UNDEFINED-ERROR? fn) UNDEFINED-ERROR]
   [(procedure? fn) (apply fn args)]
   [else NON-FUNCTION-ERROR]))
```

Combine cases

```
(define INIT-ENV
`((+ ,450+)
  (× ,450*)))
```

Check correct number of arguments???

In-class Coding 4/22: `bind` + "call" examples

```
;; A Program is one of:
;; - Atom
;; - Variable
;; - `(bind [,Var ,Program] ,Program)
;; - `(,Program . ,List<Program>)
```

Come up with your own examples!

```
(check-equal?
  (eval450 '(bind [x 10] x))
  10 ) ; no shadow
```

```
(check-equal?
  (eval450 '(bind [x 10] (bind [x 20] x))
  20 ) ; shadow
```

```
(check-equal?
  (eval450
    '(bind [x 10]
       (+ (bind [x 20]
            x)
          x)) ; 2nd x outof scope here
  30 )
```

```
(check-equal?
  (eval450
   '(bind [x 10]
      (bind [x (+ x 20)] ; x = 10 here
        x))) ; x = 30 here
   30 )
```